Application Note

Datalogger

Measuring a PRT or Thermistor without a Completion Resistor

GRANITE 6, CR6, CR1000X, CR300 Series

App Note Code: 2T-O

Revision: 10/2021

CSL I.D - 1379

RELIABLE SINCE 1974 MONITORING

CAMPBELL SCIENTIFIC

Table of contents

1. Introduction

Platinum resistance thermometers (PRT) and thermistors (temperature sensors) typically require a completion resistor or terminal input resistor (TIM) to complete the measurement circuit. Newer Campbell Scientific data loggers have built-in high precision resistors for a variety of measurement purposes. With a simple understanding of Ohm's law, these resistors can be utilized to measure an unknown resistor using PRTs.

This application note discusses two methods of measuring a PRT or thermistor without an external completion resistor. The first method uses the `CurrentSE()` instruction to make a single-ended current measurement. The second method uses the `VoltDiff()` instruction to measure an unknown resistor. The `CurrentSE()` method requires only two analogue terminals, while the `VoltDiff()` method requires four. However, the `VoltDiff()` method improves the accuracy of the measurement.

In many cases, the accuracy of the methods described in this application note are adequate. However, when high accuracy is required, using a TIM is recommended.

Table 1-1 (p. 1) shows data loggers with this capability.

Table 1-1: Campbell Scientific data loggers

Data logger	Terminals
CR300 series	SE1 and SE2, 100 Ω
CR1000X	RG1 and RG2, 101 Ω (91 Ω + 10 Ω)[1]
CR6, GRANITE 6 (serial number 8000+)	RG, 101 Ω (91 Ω + 10 Ω)[1]

[1]RG = resistive ground

2. Measuring a PRT or a thermistor using the CurrentSE() instruction

The CurrentSE() instruction is used to make a single-ended current measurement using the **SE1** and **SE2** terminals on the CR300-series data loggers, or the resistive ground (**RG**) terminals on the GRANITE 6, CR1000X, and CR6 data loggers. Using CurrentSE() will also measure the value of a resistive sensor.

2.1 CR300 series

CR300-series terminals **SE1** and **SE2** have a built-in resistor of 100 Ω to read a 4-20 mA current sensor. The accuracy of the measured current is ±0.14% of the measurement from 0 to 40 °C, or ±0.26% for the temperature range of –40 to 80 °C. This method uses the value of the excited voltage. No additional analogue channels are required.

> **NOTE:**
> Significant series-resistor errors may occur when 2-wire measurements are made with long cable lengths. In applications where high accuracy is critical, a TIM is recommended.

CRBasic Example 1 (p. 4) uses the Omega 1000 Ω PRT to demonstrate the method.

FIGURE 2-1 (p. 3) shows the wiring between the PRT/thermistor and the data logger.



FIGURE 2-1. Wiring for PRT/thermistor measurement using the `CurrentSE()` instruction with the CR300



FIGURE 2-2. CR300-series internal resistance

The current in the circuit is given as I_S. This current is flowing through a total resistance of 140 Ω built into the data logger and the resistor R_S.

The resistance R_S can be determined using following equation:

$$R_s = \frac{V_{x1}}{Is} - 140$$

CRBasic Example 1 (p. 4) demonstrates using the `CurrentSE()` instruction with a CR300-series data logger.

CRBasic Example 1: Measuring a PRT with the CR300-series data logger

```
'CR300 Series Datalogger
'Use CurrentSE() to measure an Omega 1000 Ω PRT

Public PTemp, Batt_volt, I_s, Rs, PRT_Temp

DataTable (PRTTemp,1,-1)
  DataInterval (0,15,Sec,10)
  Minimum (1,Batt_volt,FP2,False,False)
  Sample (1,PTemp,FP2)
  Sample (1,PRT_Temp,FP2)
EndTable

BeginProg
  Scan (1,Sec,0,0)
    PanelTemp (PTemp,60)
    Battery (Batt_volt)
    ExciteV (Vx1,250,0) 'Apply excitation
    CurrentSE (I_s,1,mv34,1,1,3000,60,1.0,0)
    ExciteV (Vx1,0,0) 'Turn off excitation
    Rs = (250/I_s - 140)
    PRTCalc (PRT_Temp,1,Rs/1000,1,1.0,0)
    CallTable PRTTemp
  NextScan
EndProg
```

2.2 GRANITE 6, CR6 (serial number 8000 and higher), or CR1000X

The GRANITE 6, CR6 (serial numbers 8000 and higher only), and CR1000X data loggers have a built-in resistance of 101 Ω which allows 4-20 mA sensors to be measured using an **RG** terminal. The measurement is made across a 10 Ω resistor with a 91 Ω resistor in series in the circuit. Any unknown resistor can be measured using an **RG** terminal.



FIGURE 2-3. Wiring for PRT/thermistor measurement using `CurrentSE()` *instruction with the GRANITE 6, CR6, or CR1000X*

Using `CurrentSE()` will also measure the value of a resistive sensor. In the case of a PRT or thermistor, the temperature can be measured without the use of a TIM module or any other completion resistor. FIGURE 2-4 (p. 5) shows the wiring for the GRANITE 6, CR6, and CR1000X.

The `CurrentSE()` instruction is used to measure the current In the circuit, I_S.



FIGURE 2-4. GRANITE 6, CR6, CR1000X internal resistance

The unknown sensor resistance can be calculated with this formula:

$$R_s = \frac{V_{x1}}{Is} - 101$$

The accuracy of the current measurement is about 0.1%, and the accuracy on V_{x1} is 12-bit. Together these two values will determine the accuracy of the measurement.

CRBasic Example 2 (p. 6) demonstrates measuring an Omega 1000 Ω PRT with the `CurrentSE()` instruction.

CRBasic Example 2: Measuring a PRT with the GRANITE 6, CR6, and CR1000X data loggers

```
'GRANITE 6, CR6, and CR1000X data loggers
'Use CurrentSE() to measure an Omega 1000 Ω PRT

Public PTemp, Batt_volt, I_s, Rs, PRT_Temp

DataTable (PRTTemp,1,-1)
  DataInterval (0,15,Sec,10)
  Minimum (1,Batt_volt,FP2,False,False)
  Sample (1,PTemp,FP2)
  Sample (1,PRT_Temp,FP2)
EndTable

BeginProg
  Scan (1,Sec,0,0)
    PanelTemp (PTemp,60)
    Battery (Batt_volt)
    'Change Vx1 to U terminal when using the GRANITE 6/CR6.
    ExciteV (Vx1,250,0) 'Apply excitation
    'Change the RGChan parameter to RG when using the GRANITE 6/CR6.
    CurrentSE (I_s,1,mV200,RG1,1,3000,60,1.0,0)
    'Change Vx1 to U terminal when using the GRANITE 6/CR6.
    ExciteV (Vx1,0,0) 'Turn excitation off
    Rs = (250/I_s - 101)
    PRTCalc (PRT_Temp,1,Rs/1000,1,1.0,0)
    CallTable PRTTemp
  NextScan
EndProg
```

3. Measuring an unknown resistor using the `VoltDiff()` instruction

NOTE:
This method is most relevant to the CR300-series and CR1000X data loggers. The GRANITE 6 and CR6 provide high accuracy 3-wire or 4-wire PRT measurements on native **U** terminals without a TIM. Refer to data logger specifications for details.

It is also possible to measure the voltage drop across R_S using the `VoltDiff()` instruction. This method requires an additional H and L terminal pair to measure the voltage drop across the sensor resistor. The wiring diagram is shown in FIGURE 3-1 (p. 7).



FIGURE 3-1. Wiring for PRT/thermistor measurements with extra volt diff measurement for high accuracy

The following equation can be used to determine the resistance R_S of the sensor.

$$R_s = \frac{V_1}{Is}$$

Since V_1 is measured with 24-bit accuracy, this measurement results in better accuracy compared to the `CurrentSE()` method at the cost of one differential analogue channel.

This example begins with the `CurrentSE()` instruction to determine the amperage value for I_S. It then uses the `VoltDiff()` instruction to measure the voltage value for V_1. Those variables are then used to calculate the sensor resistance value of R_S.

CRBasic Example 3: Using VoltDiff() to measure V_1

```
'GRANITE 6, CR6, CR1000X, and CR300-series data loggers
'Use VoltDiff() to measure an Omega 1000 Ω PRT

Public PTemp, Batt_volt, I_s, Rs, PRT_Temp,V1

DataTable (PRTTemp,1,-1)
  DataInterval (0,15,Sec,10)
  Minimum (1,Batt_volt,FP2,False,False)
  Sample (1,PTemp,FP2)
  Sample (1,PRT_Temp,FP2)
EndTable

BeginProg
  Scan (1,Sec,0,0)
    PanelTemp (PTemp,60)
    Battery (Batt_volt)
    'Change excitation channel to U1 (or other U terminal) when using GRANITE 6/CR6.
    ExciteV (Vx1,2500,0,0) 'Apply excitation
    'Change RGChan parameter to RG for GRANITE 6/CR6, or to SE1 or SE2 for CR300.
    CurrentSE (I_s,1,mV200,RG1,1,3000,60,1.0,0)
    'Change differential channel to U3 (or other U terminal) for GRANITE 6/CR6.
    VoltDiff (V1,1,mV1000,2,True ,3000,60,1.0,0)
    'Change excitation channel to U1 (or other U terminal) when using GRANITE 6/CR6.
    ExciteV (Vx1,0,0,0) 'Turn off excitation
    Rs = (V1/I_s)
    PRTCalc (PRT_Temp,1,Rs/1000,1,1.0,0)
    CallTable PRTTemp
  NextScan
EndProg
```

4. Measurement example

FIGURE 4-1 (p. 9) shows a measurement from two Omega 1000 Ω PRTs using the `CurrentSE()` method. One sensor was dipped in a cold bath and the other in a hot bath, and then both were removed and left in the air near each other. This measurement was made on a CR310 using VX1 as excitation for both sensors.



FIGURE 4-1. One minute average temperature measurement using CR310. Sensor on SE1 was dipped in cold water and sensor on SE2 was dipped in hot water for small period.

INFO

CAMPBELL SCIENTIFIC
WHEN MEASUREMENTS MATTER

Global Sales & Support Network

A worldwide network to help meet your needs

Campbell Scientific Regional Offices

Australia

Location: Garbutt, QLD Australia
Phone: 61.7.4401.7700
Email: info@campbellsci.com.au
Website: www.campbellsci.com.au

Brazil

Location: São Paulo, SP Brazil
Phone: 11.3732.3399
Email: vendas@campbellsci.com.br
Website: www.campbellsci.com.br

Canada

Location: Edmonton, AB Canada
Phone: 780.454.2505
Email: dataloggers@campbellsci.ca
Website: www.campbellsci.ca

China

Location: Beijing, P. R. China
Phone: 86.10.6561.0080
Email: info@campbellsci.com.cn
Website: www.campbellsci.com.cn

Costa Rica

Location: San Pedro, Costa Rica
Phone: 506.2280.1564
Email: info@campbellsci.cc
Website: www.campbellsci.cc

France

Location: Vincennes, France
Phone: 0033.0.1.56.45.15.20
Email: info@campbellsci.fr
Website: www.campbellsci.fr

Germany

Location: Bremen, Germany
Phone: 49.0.421.460974.0
Email: info@campbellsci.de
Website: www.campbellsci.de

India

Location: New Delhi, DL India
Phone: 91.11.46500481.482
Email: info@campbellsci.in
Website: www.campbellsci.in

South Africa

Location: Stellenbosch, South Africa
Phone: 27.21.8809960
Email: sales@campbellsci.co.za
Website: www.campbellsci.co.za

Spain

Location: Barcelona, Spain
Phone: 34.93.2323938
Email: info@campbellsci.es
Website: www.campbellsci.es

Thailand

Location: Bangkok, Thailand
Phone: 66.2.719.3399
Email: info@campbellsci.asia
Website: www.campbellsci.asia

UK

Location: Shepshed, Loughborough, UK
Phone: 44.0.1509.601141
Email: sales@campbellsci.co.uk
Website: www.campbellsci.co.uk

USA

Location: Logan, UT USA
Phone: 435.227.9120
Email: info@campbellsci.com
Website: www.campbellsci.com